clear all 
set more off

********************************************************************************
***** Project: The Short and Long Term Effects of In-Person Performance Feedback
********************************************************************************
***** A. R. Soetevent & G. J. Romensen
********************************************************************************
***  Eco Coach Survey Data * Data Preparation
********************************************************************************

*global filepath "C:\JPEMicReplication"
*global paperpath "$filepath\TablesGraphs"
log using "$filepath/Logs/12SurveyDataPreparation.log", replace

import excel using "$filepath\Data\Raw data\Arrivasurveyresponses01042021.xlsx", firstrow

*****************************************
**** Prepare data for analysis***
*****************************************

*** 1. Position within organisation
*** 2. Concession area
*** 3. Fuel 
*** 4. ABC

/*** 1. Position within organisation ***/

tab functie

* 1.1 Employees who answered "other: " ("Anders, namelijk:")

tab functie_anders if functie=="Anders, namelijk:"

* Teammanagers
replace functie="Teammanager" if functie_anders=="Teammanager" | functie_anders=="TM" | functie_anders=="teammanager" | functie_anders=="Teammanager "
replace functie_anders="" if functie_anders=="Teammanager" | functie_anders=="TM" | functie_anders=="teammanager" | functie_anders=="Teammanager "

tab functie_anders if functie=="Anders, namelijk:"

* Other
replace functie="Concessiemanager" if functie_anders=="Concessiemanager"
replace functie_anders="" if functie_anders=="Concessiemanager"
replace functie="ECO coordinator" if functie_anders=="Eco coördinator"
replace functie_anders="" if functie_anders=="Eco coördinator"
replace functie="Treinmachinist" if functie_anders=="Trein machinist "
replace functie_anders="" if functie_anders=="Trein machinist "

tab functie_anders if functie=="Anders, namelijk:"
* Check!

* 1.2 Split positions

gen functie_1=""
gen functie_2=""
gen functie_3=""
gen functie_4=""

tab functie

gen functie_hulp=functie if strpos(functie, "namelijk")==0

replace functie_1="ECO coach" if functie_hulp=="ECO coach,Buschauffeur"
replace functie_2="Buschauffeur" if functie_hulp=="ECO coach,Buschauffeur"

replace functie_1="ECO coach" if functie_hulp=="ECO coach,Regiodirecteur,Buschauffeur"
replace functie_2="Buschauffeur" if functie_hulp=="ECO coach,Regiodirecteur,Buschauffeur"
replace functie_3="Regiodirecteur" if functie_hulp=="ECO coach,Regiodirecteur,Buschauffeur"

replace functie_1="Locatiemanager" if functie_hulp=="Locatiemanager,Regiodirecteur"
replace functie_2="Regiodirecteur" if functie_hulp=="Locatiemanager,Regiodirecteur"

* Finalize

replace functie_1=functie if strpos(functie, "namelijk")==0 & functie_1==""

list functie functie_anders if functie_1==""

replace functie_1="ECO coach" if functie=="ECO coach,Buschauffeur,Anders, namelijk:"
replace functie_2="Buschauffeur" if functie=="ECO coach,Buschauffeur,Anders, namelijk:"

*Driver below is an eco coach
replace functie_1="ECO coach" if functie=="Buschauffeur,Anders, namelijk:"
replace functie_2="Buschauffeur" if functie=="Buschauffeur,Anders, namelijk:"

tab functie_anders

replace functie_3="Mentor" if strpos(functie_anders, "Mentor")==1
replace functie_3="Instructeur" if functie_anders=="Instructeur"
replace functie_3="Dienstindeler" if functie_anders=="Dienstindeler"
replace functie_4="B.P.V. docent" if strpos(functie_anders, "Mentor en B.P.V. docent")==1

drop functie functie_anders functie_hulp

order responseID startdate enddate progress duration finished recordeddate functie_*

/*** 2. Concession area ***/

* Based on conversation with company: Noord=Friesland, Limburg=Limburg, Zuid=Brabant, West=ZH, Oost=Gelderland

tab concessiegebied

gen region="Zuid" if concessiegebied=="Oost-Brabant" | concessiegebied=="West-Brabant"
replace region="Limburg" if concessiegebied=="Limburg"
replace region="Oost" if concessiegebied=="Achterhoek-Rivierenland"
replace region="Noord" if concessiegebied=="Noord- en Zuidwest-Fryslân en Schiermonnikoog" | concessiegebied=="Zuidoost-Fryslân en Wadden" | concessiegebied_anders=="Leeuwarden stad" | concessiegebied=="Stadsvervoer Lelystad" | concessiegebied_anders=="Lelystad"
replace region="West" if concessiegebied=="Zuid-Holland Noord"
replace region="" if concessiegebied=="Anders, namelijk:"

gen area = "FR" if concessiegebied=="Noord- en Zuidwest-Fryslân en Schiermonnikoog" | concessiegebied_anders=="Leeuwarden stad" 
replace area = "ZH" if concessiegebied=="Zuid-Holland Noord"
replace area = "OTHER" if area==""

drop concessiegebied_anders

tab region
tab area 


/*** 3. Fuel ***/

* "1 a 2 weeks" is translated into  1.5 week etc.

tab fuel_halveringweken

gen 	fuel_half=fuel_halveringweken
replace	fuel_half="0" if fuel_halveringweken=="0"
replace fuel_half="4" if fuel_halveringweken=="1 maand"
replace fuel_half="1.5" if fuel_halveringweken=="1-2 " | fuel_halveringweken=="1x pw of per 2 weken daar waar ruimte is (aantal weken)"
replace fuel_half="2" if fuel_halveringweken=="2 weken "
replace fuel_half="2.5" if fuel_halveringweken=="2 tot 3"
replace fuel_half="3" if fuel_halveringweken=="3 weken"
replace fuel_half="3.5" if fuel_halveringweken=="3 à 4 weken"
replace fuel_half="8" if fuel_halveringweken=="8 weken"
replace fuel_half="" if fuel_halveringweken=="?"


destring fuel_half, replace

/*** 4. ABC ***/

tab A_halveringweken

gen 	A_half="8"  if A_halveringweken=="2 maand"
replace A_half="2" if A_halveringweken=="2 weken"
replace A_half="4.5" if A_halveringweken=="4-5"
replace A_half="6" if A_halveringweken=="6 weken"
replace A_half="" if A_halveringweken=="N.v.t. in Brabant (geen accelaratiemeting)"

destring A_half, replace

tab B_halveringweken

gen 	B_half=B_halveringweken
replace	B_half="8" if B_halveringweken=="2 maand"
replace B_half="36" if B_halveringweken=="36weken"
replace B_half="4" if B_halveringweken=="4weken"
replace B_half="" if B_halveringweken=="Is moeilijk te zeggen, het geheel is te vrijblijvend. Weinig steun van het management op het aanspreken van chauffeurs. Dit komt ook door de rol van de OR"

destring B_half, replace

tab C_halveringweken

gen 	C_half=C_halveringweken
replace	C_half="8" if C_halveringweken=="2 maand"
replace C_half="2" if C_halveringweken=="2 weken"
replace C_half="4" if C_halveringweken=="4 weken"
replace C_half="8" if C_halveringweken=="8 weken"

destring C_half, replace

**************************************************************************************

* 1. Who filled out the survey? 
gen functiegroep=""
replace functiegroep="ECO coach en/of buschauffeur" if (functie_1=="ECO coach" & functie_2=="Buschauffeur") | (functie_1=="Buschauffeur" & functie_2=="") | (functie_1=="ECO coach" & functie_2=="") 
replace functiegroep="Teammanager" if (functie_1=="Teammanager" & functie_2=="") | (functie_1=="Locatiemanager" & functie_2=="")
replace functiegroep="Regiodirecteur" if functie_1=="Regiodirecteur" & functie_2==""
replace functiegroep="Overig" if (functie_1=="Concessiemanager" & functie_2=="") | (functie_1=="Treinmachinist" & functie_2=="") | (functie_1=="ECO coordinator" & functie_2=="") 

tab functiegroep



* Among which groups of drivers most effect is expected?
tab fuel_meestevooruitgang

gen v2fuel_meestevooruitgang=1 if fuel_meestevooruitgang=="Geen idee"
replace v2fuel_meestevooruitgang=2 if fuel_meestevooruitgang=="Alle genoemde groepen zullen nagenoeg dezelfde vooruitgang boeken"
replace v2fuel_meestevooruitgang=3 if fuel_meestevooruitgang=="Chauffeurs die tot de onderste 25% behoren op basis van brandstofverbruik"
replace v2fuel_meestevooruitgang=4 if fuel_meestevooruitgang=="Chauffeurs die tot de onderste 25%-50% behoren op basis van brandstofverbruik"
replace v2fuel_meestevooruitgang=5 if fuel_meestevooruitgang=="Chauffeurs die tot de beste 25%-50% behoren op basis van brandstofverbruik"
replace v2fuel_meestevooruitgang=6 if fuel_meestevooruitgang=="Chauffeurs die tot de beste 25% behoren op basis van brandstofverbruik"

tab ABC_meestevooruitgang

gen v2ABC_meestevooruitgang=1 if ABC_meestevooruitgang=="Geen idee"
replace v2ABC_meestevooruitgang=2 if ABC_meestevooruitgang=="Alle genoemde groepen zullen nagenoeg dezelfde vooruitgang boeken"
replace v2ABC_meestevooruitgang=3 if ABC_meestevooruitgang=="Chauffeurs die tot de onderste 25% behoren op basis van comfortabel rijden"
replace v2ABC_meestevooruitgang=4 if ABC_meestevooruitgang=="Chauffeurs die tot de onderste 25%-50% behoren op basis van comfortabel rijden"
replace v2ABC_meestevooruitgang=5 if ABC_meestevooruitgang=="Chauffeurs die tot de beste 25%-50% behoren op basis van comfortabel rijden"
replace v2ABC_meestevooruitgang=6 if ABC_meestevooruitgang=="Chauffeurs die tot de beste 25% behoren op basis van comfortabel rijden"


* Among which groups of drivers least effect is expected?
tab fuel_minstevooruitgang

gen v2fuel_minstevooruitgang=1 if fuel_minstevooruitgang=="Geen idee"
replace v2fuel_minstevooruitgang=2 if fuel_minstevooruitgang=="Alle genoemde groepen zullen nagenoeg dezelfde vooruitgang boeken"
replace v2fuel_minstevooruitgang=3 if fuel_minstevooruitgang=="Chauffeurs die tot de onderste 25% behoren op basis van brandstofverbruik"
replace v2fuel_minstevooruitgang=4 if fuel_minstevooruitgang=="Chauffeurs die tot de onderste 25%-50% behoren op basis van brandstofverbruik"
replace v2fuel_minstevooruitgang=5 if fuel_minstevooruitgang=="Chauffeurs die tot de beste 25%-50% behoren op basis van brandstofverbruik"
replace v2fuel_minstevooruitgang=6 if fuel_minstevooruitgang=="Chauffeurs die tot de beste 25% behoren op basis van brandstofverbruik"

tab ABC_minstevooruitgang

gen v2ABC_minstevooruitgang=1 if ABC_minstevooruitgang=="Geen idee"
replace v2ABC_minstevooruitgang=2 if ABC_minstevooruitgang=="Alle genoemde groepen zullen nagenoeg dezelfde vooruitgang boeken"
replace v2ABC_minstevooruitgang=3 if ABC_minstevooruitgang=="Chauffeurs die tot de onderste 25% behoren op basis van comfortabel rijden"
replace v2ABC_minstevooruitgang=4 if ABC_minstevooruitgang=="Chauffeurs die tot de onderste 25%-50% behoren op basis van comfortabel rijden"
replace v2ABC_minstevooruitgang=5 if ABC_minstevooruitgang=="Chauffeurs die tot de beste 25%-50% behoren op basis van comfortabel rijden"
replace v2ABC_minstevooruitgang=6 if ABC_minstevooruitgang=="Chauffeurs die tot de beste 25% behoren op basis van comfortabel rijden"

* Repetition coaching sessions necessary? 
tab fuel_herhalingsessie

gen v2fuel_herhalingsessie=1 if fuel_herhalingsessie=="Ja"
replace v2fuel_herhalingsessie=2 if fuel_herhalingsessie=="Nee, eenmaal is voldoende"
replace v2fuel_herhalingsessie=3 if fuel_herhalingsessie=="Nee, ook bij herhaaldelijk coachen zal het effect minimaal zijn"
replace v2fuel_herhalingsessie=4 if fuel_herhalingsessie=="Nee, herhaaldelijk coachen is niet wenselijk omdat het effect negatief is"

tab ABC_blijvend
gen v2ABC_blijvend=1 if ABC_blijvend=="Ja"
replace v2ABC_blijvend=2 if ABC_blijvend=="Nee, eenmaal is voldoende"
replace v2ABC_blijvend=3 if ABC_blijvend=="Nee, ook bij herhaaldelijk coachen zal het effect minimaal zijn"
replace v2ABC_blijvend=4 if ABC_blijvend=="Nee, herhaaldelijk coachen is niet wenselijk omdat het effect negatief is"


* 3. ABC 

local ABC "fuel A B C"


foreach var of local ABC {
gen `var'_halvering=.
replace `var'_halvering=1 if `var'_half==0
replace `var'_halvering=2 if `var'_half>=1 & `var'_half<=3
replace `var'_halvering=3 if `var'_half>3 & `var'_half<=6
replace `var'_halvering=4 if `var'_half>6 & `var'_half<=9
replace `var'_halvering=5 if `var'_half>9 & `var'_half<=12
replace `var'_halvering=6 if `var'_half>12 & `var'_half!=.

* Comparison ECO coaches vs. teammanagers
* Lasting impact?
tab `var'_blijvend
tab `var'_blijvend functiegroep

* In case not lasting, how long till effect is halved?
tab `var'_half
tab `var'_halvering
sum `var'_halvering, detail
tab `var'_halvering functiegroep

}



/*** Anonymize data ***/

drop startdate - recordeddate functie_1- functie_4 suggestie

/*** Save the data for analysis ***/

save "$filepath\DEPO\SurveyDataDEPO.dta", replace

log close
